Partner
Ecosystem
Success

Exercise 2 – Build a custom API for side-by-side extensibility

S/4HANA Cloud Developer Extensibility bootcamp

TABLE OF CONTENTS

www.sap.com/contactsap

THE BEST RUN

Introduction

With Developer Extensibility, it is now possible to create custom APIs in your S/4HANA Cloud system. These can be both inbound and outbound APIs and allow you to establish communication with external systems. As a developer, you use ABAP Development Tools for Eclipse (ADT) to implement such APIs. Keep in mind that whenever you are communicating with an external system, be it inbound or outbound communication, an administrator needs to perform the corresponding communication management to enable the usage of the desired APIs.

When developing custom **inbound** APIs, you have the option of using OData, plain HTTP or RFC communication.

Such custom APIs can be called remotely, for example from SAP Business Technology Platform. In a purely ABAP-based side-by-side extensibility scenario, your custom S/4HANA Cloud API could be consumed from an ABAP Cloud system on SAP BTP (i.e. SAP BTP, ABAP Environment).

Exercise scope

In this exercise, we will create a custom inbound OData API in an S/4HANA Cloud system. We will reuse the business object from the first day, i.e., the online shop, and create an OData service binding of type Web API on top of it. We will also create a custom communication scenario to expose our API.

In a real scenario, the development would be performed in the development tenant (client 080), while the communication management would be performed in the customizing tenant (client 100). This is because that is where we would normally call our custom API, in the customizing tenant containing more complete business configuration and data. For the purposes of this exercise, we will perform all activities in the development tenant.

We will consume the API from an ABAP system running on SAP BTP, i.e., an SAP BTP ABAP Environment service instance. We will use the API's metadata to create a service consumption model, which will allow us to call the API directly from ABAP coding. As a simple example, we will consume the API from a plain HTTP service, which will contain the custom coding in its handler class.

This HTTP service is a form of outbound communication, and thus also requires some communication management. We will therefore create another custom communication scenario to expose our outbound service.

Note:

A service consumption model analyzes an API's specifications and generates the ABAP development objects required to model the API. These allow for straightforward consumption of the API using ABAP coding. The service consumption model object page in ADT offers some code snippets for the different operations supported by the API. You can use these as a starting point for your own implementations.

Pre-requisites

ADT is required, as in Exercise 1. We will reuse the business object from the first exercise, so make sure that all objects up to the service definition are active and properly implemented.

Recommendations

When naming objects, please make sure to include your ID in the name. This will uniquely identify your objects and prevent any confusion.
Example : Z_SHOP_API_<ID>

STEP 1 – CREATE API

In S/4HANA Cloud, create and expose an inbound OData API

Step 1.1 – Create inbound API

As a developer, implement an OData service binding of type ‘Web API’ and create all required communication management development objects.

Description	Screenshot
Login to the S/4HANA Cloud Development Tenant via Eclipse To do this open Eclipse using the icon in the Remote Desktop.	
Create a new workspace and make sure you use your user folder. If the folder does not already exist, create a new one. Example: C:\Users\STUDENT<XXX>\eclipse-workspace where <XXX> is your ID (e.g. D01).	

Description	Screenshot
Open the ABAP Perspective by clicking the “Open Perspective” icon on the top right and then selecting “ABAP” and finally pressing “Open”.	
On the left side you have the Project Explorer bar. Click on “Create an ABAP Cloud Project”	
Select the first option “SAP S/4HANA Cloud ABAP Environment”. Use the S/4HANA Cloud URL available in the Cheat Sheet. This is connecting you to the Development Tenant. Make sure there are no spaces in between.	

Description	Screenshot
Open the system login page using “Open Logon Page in Browser”. Use your credentials from system “A” in your Cheat Sheet.	
Once the login is done successfully, you can close the browser window and you should see a success message in Eclipse “Logged on….”. Click “Next”.	

Description	Screenshot
Define the local project name. Use your user ID as a prefix to distinguish your project. Click “Finish”.	
Select File >> New >> ABAP Package	

Description	Screenshot
Create a new package for this exercise	
Create a new *Service Binding* object of type *Web API* Choose the service definition created in earlier exercises as your referenced object In this exercise we are using OData V2, but you could also use OData V4. Activate and publish the service binding.	
Make sure that a *COM Inbound Service* object was created under *Cloud Communication Management*. This is auto generated.	
Create a custom *Communication Scenario* object Z_COM_SHOP_SCENARIO_XXX Provide description: Scenario for inbound calls to online shop API Set *Allowed Instances* to one instance per client, as we only want to set up one communication arrangement.	

Description	Screenshot
In the *Inbound* tab, add the Inbound Service from the earlier step to your custom communication scenario. Make sure that Basic Authentication is enabled, as that is what we are going to use (to keep things simple). Save the *Communication Scenario.* Go back to the overview tab and Publish Locally.	

Step 1.2 – Expose inbound API

As an administrator, perform the required communication management to expose the inbound API.

Description	Screenshot
Login to the S/4HANA Cloud Development Tenant (080) Launchpad via your browser. You can find the link in the cheat sheet.	
Open the *Communication Arrangements* Fiori app. Create a new communication arrangement, using the previously created custom *Communication Scenario* as your reference. **Please do not modify any existing communication arrangements or other artifacts that do not belong to you!**	

Press the *New* button next to the *Communication System* field. Choose a System ID and a System Name. After pressing *Create*, you will be forwarded to the configuration page of your new communication system. Choose a system ID and system name.	
Check the *Inbound Only* checkbox under *Technical Data – General.* Since we are only exposing an inbound API, we do not need to specify any host name or such.	
Under *Users for Inbound Communication*, add a new technical user via the **+** button. Since we do not yet have a suitable user, we can create one directly from the current screen using the *New User* button.	
Specify a user name and a description. Specify a password or generate one using the *Propose Password* button. Make sure to store the password, as we will need it later. Press *Create* to save your settings. **Note**: Communication users can authenticate via password or via certificate. We will be using password authentication in this exercise.	

As you navigate back to the communication system, confirm that you want to add the newly created communication user by pressing *OK*	
Save your communication system. You will be redirected back to the original communication arrangement.	
Save your communication arrangement. The previously created communication user will automatically be used. If you were to maintain multiple users in your communication system, you could select one at the arrangement level.	
Before we move on, let's download the service metadata of the inbound service. Retrieve the metadata as follows: 1. Copy the *Service URL* shown in your arrangement 2. Remove the *"-api"* substring 3. Append "*/$metadata"* to the URL 4. Open the resulting URL in a browser 5. Save the displayed file using the context menu → *Save As…* Example (Actual URL can be obtained as per above steps): https://my400788.lab.s4hana.cloud.sap/sap/opu/odata/sap/Z_SHOP_API_WEB_V2_XXX/$metadata If there is an option to download the service metadata directly from the communication arrangement app, you could also do this.	

This metadata file will allow us to comfortably consume the service from SAP BTP. Make sure to note down the Service URL of your inbound service as well, as we will need it at a later point.	

STEP 2 – CONSUME API

Consume your custom OData API in SAP BTP ABAP Environment

Step 2.1 – Create Service Consumption Model

As a developer on SAP BTP, implement a service consumption model to consume the external OData API.

Description	Screenshot
Login to the SAP BTP ABAP Environment instance via Eclipse. Start by creating a new ABAP Cloud Project and choosing the option *SAP BTP ABAP Environment – Use a Service Key*	
Paste in the service key of the ABAP Environment instance. You can find it at the end of this document HERE.	

Description	Screenshot
Proceed to the next step and choose *Open Logon Page in Browser*. Select the second option called “httpsalvn9xegs….” Enter the credentials from your cheat sheet. Once logged in successfully close the browser window. Follow the remaining steps of the wizard.	
Select File >> New >> ABAP Package	
Create a new package for this exercise under structure package ZLOCAL.	

Description	Screenshot
Create a new *Service Consumption Model* object. Specify a name, a description and choose *OData* as the *Remote Consumption Mode*. Press *Next*.	
Locate the metadata file that you previously downloaded. Use it to define the service consumption model. Press *Next*.	

Description	Screenshot
The creation wizard will show you a list of the entities defined in the service. Make sure that the relevant entities are selected. In this case we just have the one entry. Press *Next*. The wizard will show you a list of the objects that shall be generated. You can then use these objects to consume the external API directly from your ABAP code. **Hint: If the generated ABAP Artifact Name is not easy to read, you could edit this to provide a unique name** You can now proceed with the creation.	
Make sure to activate your *Service Consumption Model* once it is created.	

Step 2.2 – Create HTTP service

As a developer on SAP BTP, implement a plain HTTP service which consumes the previously created service consumption model.

Description	Screenshot
Create a new *HTTP Service* object. Choose a name and a description, as well as a name for the handler class. The class will be automatically generated. Save and activate.	
Create a new *Outbound Service* object of type *HTTP Service*.	
Specify the default path prefix for your service. You can get this information from the communication arrangement on S/4HANA Cloud side. Example: /sap/opu/odata/sap/<binding-name> Save the object.	

Description	Screenshot
Create a custom *Communication Scenario* object.	
Set *Allowed Instances* to one instance per client, as we only want to set up one communication arrangement.	
Under *Outbound*, add the previously created *Outbound Service.* Make sure that *Basic Authentication* is enabled. Save and publish the scenario locally.	

Description	Screenshot
Now we will implement the HTTP service. Navigate to the generated handler class, where we will start to implement the `if_http_service_extension~handle_request` method. The name of the handler class was specified when we created the HTTP service in a previous step. We can get a good starting point by using the code snippets provided in the service consumption model object. Select the *Read* operation and copy the provided code over into your method.	

Replace instances of <structure_name> with the name of the CDS view (Data Definition) that was generated as part of the consumption model.
This name is shown on the service consumption model object page
	`METHOD if_http_service_extension~handle_request.`
	
	`DATA:`
	`ls_entity_key TYPE <structure_name>,`
	`ls_business_data TYPE <structure_name>,`
	`lo_http_client TYPE REF TO if_web_http_client,`
	`lo_resource TYPE REF TO /iwbep/if_cp_resource_entity,`
	`lo_client_proxy TYPE REF TO /iwbep/if_cp_client_proxy,`
	`lo_request TYPE REF TO /iwbep/if_cp_request_read,`
	`lo_response TYPE REF TO /iwbep/if_cp_response_read.`

Uncomment the method call `cl_http_destination_provider=>create_by_comm_arrangement` and adjust the parameters:
 	
- Specify the name of the custom *Communication Scenario* that you just created
- Specify the name of the *Outbound Service* that you just created
- We do not need to specify a communication system
 	
Using this information, the correct service endpoint will be derived at runtime, provided that a corresponding communication arrangement and system are maintained (see later steps).
 	
Uncomment the `cl_web_http_client_manager=>create_by_http_destination` method call.
	`TRY.`
	`" Create http client`
	`" Details depend on your connection settings`
	`DATA(lo_destination) = cl_http_destination_provider=>create_by_comm_arrangement(`
	`comm_scenario = 'Z_SHOP_SCENARIO_OUTBOUND_XXX'`
	`service_id = 'Z_SHOP_API_READ_OBS_XXX_REST' ).`
	
	`lo_http_client = cl_web_http_client_manager=>create_by_http_destination( lo_destination ).`

Replace the string `<service_root>` with an empty string. We have already specified the path in the outbound service object, hence there is no need to maintain it here.	`lo_client_proxy = cl_web_odata_client_factory=>create_v2_remote_proxy(`
	`EXPORTING`
	`iv_service_definition_name = 'Z_SHOP_API_SCM_XXX'`
	`io_http_client = lo_http_client`
	`iv_relative_service_root = '' ).`

Description	Screenshot
Specify the key of the entry that you want to read. For this exercise, we will simply hardcode a single (existing) value. You can view the UUIDs of all entries by opening the underlying database table (from the first exercise) in your S/4HANA Cloud project and using the *Data Preview* functionality.	

Use the results from the read operation to pass some meaningful information to the end user.
 	
We will write the Order ID and the generated purchase requisition ID to the response body using the appropriate setter method.
	`DATA: result type string.`
	`result = | OrderID: { ls_business_data-Order_Id },`
	`Purchase Requisition ID: { ls_business_data-`
	`Purchasereqn } |.`
	`response->set_text( result ).`

Description	Screenshot
Catch exceptions `cx_http_dest_provider_error` and `cx_web_http_client_error`. Make sure that all caught exceptions write something descriptive to the response text.	CATCH /iwbep/cx_cp_remote INTO DATA(lx_remote). " Handle remote Exception " It contains details about the problems of your http(s) connection response->set_text(| Remote Error: { lx_remote->get_longtext( ) } |). CATCH /iwbep/cx_gateway INTO DATA(lx_gateway). " Handle Exception response->set_text(| Gateway Error: { lx_gateway->get_longtext( ) } |). CATCH cx_http_dest_provider_error INTO DATA(lx_destination). "handle exception response->set_text(| Destination Error: { lx_destination->get_longtext( ) } |). CATCH cx_web_http_client_error INTO DATA(lx_http). "handle exception response->set_text(| HTTP Error: { lx_http->get_longtext( ) } |).
Optional If you want, you can also create additional HTTP services to perform different operations. For example, you can create another service to perform a *Create* operation. You will need to implement a handler class for each HTTP service. You can use the different code snippets provided in the service consumption model. Remember to create additional *Outbound Services* for each additional HTTP service and add them to your custom communication scenario.	

Step 2.3 – Expose HTTP service

As an administrator on SAP BTP, perform the outbound communication management required to consume the external OData API.

Description	Screenshot
Login to the SAP BTP ABAP Environment instance's Launchpad via your browser. You can find the link in the Cheat Sheet. Login with the user and password from your cheat sheet.	
Open the *Communication Arrangements* Fiori app Create a new communication arrangement, using the previously created custom *Communication Scenario* as your reference	
As before, press the *New* button next to the *Communication System* field. Choose a System ID and a System Name. After pressing *Create*, you will be forwarded to the configuration page of your new communication system.	
Under *Host Name*, add the host name of the S/4HANA Cloud Customizing tenant. You can read the value from the communication arrangement on S/4HANA Cloud side (*API-URL* field).	

Description	Screenshot
Under *Users for Outbound Communication*, add a new technical user via the **+** button. As we are using Basic Authentication, you need to choose the authentication method *User Name and Password*. You will need to specify the same user name and password that you configured on S/4HANA Cloud side. Press *Create*.	
Save your communication system. You will be redirected back to the original communication arrangement.	
Save your communication arrangement. The previously created communication user will automatically be used. You are now ready to consume your custom API!	

Step 2.4 – Consume API

As a developer on SAP BTP, retrieve the URL of your HTTP service.

As a business user, call your custom HTTP service.

Description	Screenshot
Login to the SAP BTP ABAP Environment instance via Eclipse	
Open the HTTP service object. Under *URL*, you will see the endpoint of your service. Copy it into your browser or use the available hyperlink to open it directly from Eclipse.	
Open the URL in a browser window or click directly on the hyperlink in ABAP Development Tools. Authenticate, if required. Check that the desired data is shown in your browser.	

Notes

- Your user will automatically have authorization for your custom HTTP service. This is because the Developer business role, which is assigned to your user, automatically grants authorization for all such custom services. To enable its usage for other business users, you will need to perform the necessary Identity and Access management for the service. This is outside the scope of this exercise.
- You can debug the remote call by setting a breakpoint in Eclipse (double-click to the left of the line number marker). We recommend that you set one at the beginning of the `if_http_service_extension~handle_request` method and follow the execution step-by-step.
- In the course of the exercise, we have used some useful ABAP features in our SAP BTP ABAP Environment instance: service consumption models and custom HTTP services. Since S/4HANA Cloud shares the same technology stack, these same features are also available in S/4HANA Cloud Developer Extensibility.

RESOURCES

SAP BTP ABAP Environment – Service Key


```
{
  "binding": {
    "env": "cf",
    "id": "8b09dcaa-d3c3-42f8-b35f-520efc2d818e",
    "type": "oauth",
    "version": "1.0.1.1"
  },
  "catalogs": {
    "abap": {
      "path": "/sap/opu/odata/IWFND/CATALOGSERVICE;v=2",
      "type": "sap_abap_catalog_v1"
    }
  },
  "endpoints": {
    "abap": "https://7e04e291-8515-4d2b-a418-f2868669f491.abap.eu10.hana.ondemand.com"
  },
  "preserve_host_header": true,
  "sap.cloud.service": "com.sap.cloud.abap",
  "systemid": "H01",
  "uaa": {
    "apiurl": "https://api.authentication.eu10.hana.ondemand.com",
    "clientid": "sb-xs-7e04e291-8515-4d2b-a418-f2868669f491!b175432|xsuaa-abapcp-prod-eu10!b4584",
    "clientsecret": "Mo8WI29fWPOFkWYNu+5At3YicSA=",
    "credential-type": "instance-secret",
    "identityzone": "abap-steampunk-v3-2tlbdp49",
    "identityzoneid": "0b5d473f-0308-43e3-945f-a61793af34fb",
    "sburl": "https://internal-xsuaa.authentication.eu10.hana.ondemand.com",
    "subaccountid": "0b5d473f-0308-43e3-945f-a61793af34fb",
    "tenantid": "0b5d473f-0308-43e3-945f-a61793af34fb",
    "tenantmode": "dedicated",
    "uaadomain": "authentication.eu10.hana.ondemand.com",
    "url": "https://abap-steampunk-v3-2tlbdp49.authentication.eu10.hana.ondemand.com",
    "verificationkey": "-----BEGIN PUBLIC KEY-----
\nMIIBIjANBgkqhkiG9w0BAQEFAAOCAQ8AMIIBCgKCAQEApwb5GK+uj0E0bn2ImZdJ\nePR7fxWp2kvG9YtmjLP3veddOcYzKc9
J2VlPWQmCyIlXh7DWMLKABylvNCcz8g0a\nHrRtHHdZkFJEHJt/MwyKIMTirjFjHAcNdIXcQYuLJAfrjNhElztfv3sWwVCfvPlC\nHrQ
2Nt8vKFuLUlu6pgKuf2e1fWtNgmxpHlae04H1t9GjoqFCJyp9zy4sk2l3PNlG\nmp01ca5RPpnZ3G9lVCTE6xD/Sl5h8CeG2kKma4og
cAQdP2UThQh8TUfQGRe4g2A+\nUunQ5R5oP3DO+PS+4TTTkqSivTXbeLcmBLe7s56JvJHM6oaJwjI3P70PrqFPCo7B\ndQIDAQ
AB\n-----END PUBLIC KEY-----",
    "xsappname": "xs-7e04e291-8515-4d2b-a418-f2868669f491!b175432|xsuaa-abapcp-prod-eu10!b4584",
    "zoneid": "0b5d473f-0308-43e3-945f-a61793af34fb"
  },
  "url": "https://7e04e291-8515-4d2b-a418-f2868669f491.abap.eu10.hana.ondemand.com"
}
```

Sample Code
Please find here some sample code for the HTTP service's handler class:

```
CLASS zcl_shop_api_read_xxx DEFINITION
  PUBLIC
  CREATE PUBLIC .

  PUBLIC SECTION.

    INTERFACES if_http_service_extension .
  PROTECTED SECTION.
  PRIVATE SECTION.
ENDCLASS.



CLASS zcl_shop_api_read_xxx IMPLEMENTATION.


  METHOD if_http_service_extension~handle_request.

    DATA:
      ls_entity_key    TYPE zonline_shop,
      ls_business_data TYPE zonline_shop,
      lo_http_client   TYPE REF TO if_web_http_client,
      lo_resource      TYPE REF TO /iwbep/if_cp_resource_entity,
      lo_client_proxy  TYPE REF TO /iwbep/if_cp_client_proxy,
      lo_request       TYPE REF TO /iwbep/if_cp_request_read,
      lo_response      TYPE REF TO /iwbep/if_cp_response_read.



    TRY.
        " Create http client
        DATA(lo_destination) = cl_http_destination_provider=>create_by_comm_arrangement(
                                                 comm_scenario  = 'Z_SHOP_SCENARIO_OUTBOUND_XXX'
*                                            comm_system_id = '<Comm System Id>'
                                                 service_id     = 'Z_SHOP_API_READ_OBS_XXX_REST' ).
        lo_http_client = cl_web_http_client_manager=>create_by_http_destination( lo_destination ).

        lo_client_proxy = cl_web_odata_client_factory=>create_v2_remote_proxy(
          EXPORTING
            iv_service_definition_name = 'Z_SHOP_API_SCM_XXX'
            io_http_client             = lo_http_client
            iv_relative_service_root   = '' ).


        " Set entity key
        ls_entity_key = VALUE #(
                  order_uuid  = '5B3F46742F341EEDA5C9E32C19BEDA01' ).

        " Navigate to the resource
        lo_resource = lo_client_proxy->create_resource_for_entity_set( 'ONLINE_SHOP' )->navigate_with_key(
ls_entity_key ).

        " Execute the request and retrieve the business data
        lo_response = lo_resource->create_request_for_read( )->execute( ).
        lo_response->get_business_data( IMPORTING es_business_data = ls_business_data ).

        DATA: result TYPE string.
        result = | OrderID: { ls_business_data-Order_Id },  Purchase Requisition ID: { ls_business_data-
Purchasereqn  } |.
        response->set_text( result ).


      CATCH /iwbep/cx_cp_remote INTO DATA(lx_remote).
        " Handle remote Exception
        " It contains details about the problems of your http(s) connection
        response->set_text( | Remote Error: { lx_remote->get_longtext(  ) } | ).
      CATCH /iwbep/cx_gateway INTO DATA(lx_gateway).
        " Handle Exception
        response->set_text( | Gateway Error: { lx_gateway->get_longtext(  ) } | ).
      CATCH cx_http_dest_provider_error INTO DATA(lx_destination).
        "handle exception
```

```
        response->set_text( | Destination Error: { lx_destination->get_longtext(  ) } | ).
      CATCH cx_web_http_client_error INTO DATA(lx_http).
        "handle exception
        response->set_text( | HTTP Error: { lx_http->get_longtext(  ) } | ).
    ENDTRY.
  ENDMETHOD.
ENDCLASS.
```